<commit_before>{ lib, stdenv, fetchFromGitHub, autoconf, automake, libtool, pkg-config
, libxml2, nettle
, withGTK3 ? true, gtk3 }:

stdenv.mkDerivation rec {
  pname = "stoken";
  version = "0.92";
  src = fetchFromGitHub {
    owner = "cernekee";
    repo = pname;
    rev = "v${version}";
    sha256 = "0q7cv8vy5b2cslm57maqb6jsm7s4rwacjyv6gplwp26yhm38hw7y";
  };

  preConfigure = ''
    aclocal
    libtoolize --automake --copy
    autoheader
    automake --add-missing --copy
    autoconf
  '';

  nativeBuildInputs = [ pkg-config ];
  buildInputs = [
    autoconf automake libtool
    libxml2 nettle
  ] ++ lib.optional withGTK3 gtk3;

  meta = with lib; {
    description = "Software Token for Linux/UNIX";
    homepage = "https://github.com/cernekee/stoken";
    license = licenses.lgpl21Plus;
    maintainers = [ ];
    platforms = platforms.all;
  };
}<commit_msg>stoken: fix cross compilation<commit_after>{ lib, stdenv, fetchFromGitHub, autoconf, automake, libtool, pkg-config
, libxml2, nettle
, withGTK3 ? true, gtk3 }:

stdenv.mkDerivation rec {
  pname = "stoken";
  version = "0.92";
  src = fetchFromGitHub {
    owner = "cernekee";
    repo = pname;
    rev = "v${version}";
    sha256 = "0q7cv8vy5b2cslm57maqb6jsm7s4rwacjyv6gplwp26yhm38hw7y";
  };

  preConfigure = ''
    aclocal
    libtoolize --automake --copy
    autoheader
    automake --add-missing --copy
    autoconf
  '';

  strictDeps = true;
  nativeBuildInputs = [ pkg-config autoconf automake libtool ];
  buildInputs = [
    libxml2 nettle
  ] ++ lib.optional withGTK3 gtk3;

  meta = with lib; {
    description = "Software Token for Linux/UNIX";
    homepage = "https://github.com/cernekee/stoken";
    license = licenses.lgpl21Plus;
    maintainers = [ ];
    platforms = platforms.all;
  };
}
{'fraction_non_alphanumeric': 0.12296462661426165, 'fraction_numerical': 0.033688938798427846, 'mean_word_length': 3.1733021077283374, 'pattern_counts': {'":': 0, '<': 3, '<?xml version=': 0, '>': 3, 'https://': 2, 'lorem ipsum': 0, 'www.': 0, 'xml': 4}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '6598441', 'n_tokens_mistral': 702, 'n_tokens_neox': 674, 'n_words': 135}